



**Министерство науки и высшего образования Российской Федерации
Федеральное государственное бюджетное
образовательное учреждение высшего образования
«Московский государственный технический университет
имени Н.Э. Баумана
(национальный исследовательский университет)»
(МГТУ им. Н.Э. Баумана)**

ФАКУЛЬТЕТ Информатика и системы управления

КАФЕДРА Программное обеспечение ЭВМ и информационные технологии

Отчет по лабораторной работе №7

По курсу “Моделирование”

Тема: “Определение вероятности отказа заданной системы”

Студент Лучина Е.Д (№ в списке 13)

Группа ИУ7-71Б

Преподаватель Рудаков И.В.

Москва

2020 г.

Задание

В информационный центр приходят клиенты через интервалы времени 10 ± 2 минуты. Если все три имеющихся оператора заняты клиенту отказывают в обслуживании. Операторы имеют разную производительность и могут обеспечивать обслуживание среднего запроса пользователей за 20 ± 5 , 40 ± 10 , 40 ± 20 минут. Клиенты стремятся занять свободного оператора с максимальной производительностью. Полученные запросы сдаются в приемный накопитель откуда и выбираются на обработку: на первый компьютер запрос от первого и второго операторов, на второй от третьего оператора. Время обработки запросов на первом и втором компьютере равны 15 и 30 минут соответственно.

Промоделировать процесс обработки 300 запросов, оценить вероятность отказа. Необходимо для этого создать концептуальную модель в терминах СМО, определить эндогенные и экзогенные переменные и уравнения модели. За единицу системного времени выбрать 0,01 минуты.




Рисунок 1. - схема системы

Модель системы имеет следующие составляющие (рисунок 1):

- блок имитатора воздействия внешней среды:
 - Генератор заявок (каждые 10 ± 2 минуты)
- блок функций системы:
 - Три оператора (первостепенная обработка заявок; Клиенты стремятся занять свободного оператора с максимальной производительностью.)
 - Оператор 1 (20 ± 5 минут на обработку одной заявки)
 - Оператор 2 (40 ± 10 минут на обработку одной заявки)
 - Оператор 3 (40 ± 20 минут на обработку одной заявки)
 - Два накопителя (две буферные памяти)
 - Накопитель 1 (приходят заявки от первого и второго оператора)
 - Накопитель 2 (приходят заявки от третьего оператора)
 - Два компьютера (второстепенная обработка заявки)
 - Компьютер 1 (время обработки запроса составляет 15 минут)
 - Компьютер 2 (время обработки запроса составляет 30 минут)
- вспомогательный блок:
 - Управляющая программа

Эндогенные (зависимые) переменные модели – время обработки i -ым оператором, и время решения на j -ом компьютере

Экзогенные переменные: n_0 – число обслуженных клиентов, n_1 – число клиентов получивших отказываю. Чтобы оценить вероятность отказа P (в чем и заключается поставленная задача) воспользуемся следующей формулой (1):

$$P = \frac{n_1}{(n_0 + n_1)} \quad (1)$$

Программа на GPSS приведена в листинге 1. Отчет моделирования прикреплен к лабораторной работе в виде файлов rfd и grg (ввиду проблем форматирования). Вероятность отказа в обслуживании составляет 23 %.

Листинг 1- программа на GPSS

```

SIMULATE
; В информационный центр приходят максимум 300 клиентов через интервалы времени 10 ± 2 секунды
GENERATE 10,2,,300,

; Система имеет 3 оператора. Если все операторы заняты клиенту отказывают в обслуживании.
; Операторы обеспечивают обслуживание среднего запроса за 20 ± 5, 40 ± 10, 40 ± 20 секунды.
; Клиенты стремятся занять свободного оператора с максимальной производительностью.
; Полученные запросы подаются в приемный накопитель откуда и выбираются на обработку на компьютеры
; на первый компьютер запрос от первого и второго операторов, на второй от третьего оператора

OP_1  GATE NU OP_1_USE,OP_2          ; переход ко второму оператору, если первый занят
      SEIZE OP_1_USE                  ; клиент занимает устройство
      ADVANCE 20,5                     ; использование устройства 20 ± 5 секунд
      RELEASE OP_1_USE                ; освобождение устройства
      TRANSFER ,PC_1,,                ; переход на обработку в первый компьютер

OP_2  GATE NU OP_2_USE,OP_3          ; переход к третьему оператору, если второй занят
      SEIZE OP_2_USE                  ; клиент занимает устройство
      ADVANCE 40,10                    ; использование устройства 40 ± 10 секунд
      RELEASE OP_2_USE                ; освобождение устройства
      TRANSFER ,PC_1,,                ; переход на обработку в первый компьютер

OP_3  GATE NU OP_3_USE,REJECTED      ; переход к блоку отказа, если третий оператор занят
      SEIZE OP_3_USE                  ; клиент занимает устройство
      ADVANCE 40,20                    ; использование устройства 40 ± 20 секунд
      RELEASE OP_3_USE                ; освобождение устройства
      TRANSFER ,PC_2,,                ; переход на обработку во второй компьютер

; Система имеет два компьютера, каждый из которых имеет приемный накопитель.
; Время обработки запросов на первом и втором компьютере равны 15 и 30 минут соответственно.

PC_1  QUEUE Q_PC_1                   ; добавление клиента в очередь PC1
      SEIZE PC_1_USE                   ; клиент занимает устройство
      DEPART Q_PC_1                   ; излечение клиента из очереди
      ADVANCE 15,                      ; использование устройства 15 секунд
      RELEASE PC_1_USE                ; освобождение устройства
      TRANSFER ,PROCESSED,,            ; клиент обработан, переход в блок processed

PC_2  QUEUE Q_PC_2                   ; добавление клиента в очередь PC2
      SEIZE PC_2_USE                   ; клиент занимает устройство
      DEPART Q_PC_2                   ; излечение клиента из очереди
      ADVANCE 30,                      ; использование устройства 30 секунд
      RELEASE PC_2_USE                ; освобождение устройства
      TRANSFER ,PROCESSED,,            ; клиент обработан, переход в блок processed

; Обработанные заявки
PROCESSED TRANSFER ,RES,,              ; переход к подсчету переменных

; Отказанный заявки
REJECTED TRANSFER ,RES,,              ; переход к подсчету переменных

```

; Подсчет необходимых переменных

RES SAVEVALUE VAR_COUNT_PROC,N\$PROCESSED ; количество обработанных заявок
 SAVEVALUE VAR_COUNT_REJECT,N\$REJECTED ; количество отказов
 SAVEVALUE VAR_PROBABILITY,((N\$REJECTED) / (N\$RES)) ; вероятность отказа

TERMINATE 1

START 300